// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/math/Math.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/math/Math.sol)

pragma solidity ^0.8.0;

/**
 * @dev Standard math utilities missing in the Solidity language.
 */
library Math {
    enum Rounding {
        Down, // Toward negative infinity
        Up, // Toward infinity
        Zero // Toward zero
    }

    /**
     * @dev Returns the largest of two numbers.
     */
    function max(uint256 a, uint256 b) internal pure returns (uint256) {
        return a > b ? a : b;
    }

    /**
     * @dev Returns the smallest of two numbers.
     */
    function min(uint256 a, uint256 b) internal pure returns (uint256) {
        return a < b ? a : b;
    }

    /**
     * @dev Returns the average of two numbers. The result is rounded towards
     * zero.
     */
    function average(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b) / 2 can overflow.
        return (a & b) + (a ^ b) / 2;
    }

    /**
     * @dev Returns the ceiling of the division of two numbers.
     *
     * This differs from standard division with `/` in that it rounds up instead
     * of rounding down.
     */
    function ceilDiv(uint256 a, uint256 b) internal pure returns (uint256) {
        // (a + b - 1) / b can overflow on addition, so we distribute.
        return a == 0 ? 0 : (a - 1) / b + 1;
    }

    /**
     * @notice Calculates floor(x * y / denominator) with full precision. Throws if result overflows a uint256 or denominator == 0
     * @dev Original credit to Remco Bloemen under MIT license (https://xn--2-umb.com/21/muldiv)
     * with further edits by Uniswap Labs also under MIT license.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator
    ) internal pure returns (uint256 result) {
        unchecked {
            // 512-bit multiply [prod1 prod0] = x * y. Compute the product mod 2^256 and mod 2^256 - 1, then use
            // use the Chinese Remainder Theorem to reconstruct the 512 bit result. The result is stored in two 256
            // variables such that product = prod1 * 2^256 + prod0.
            uint256 prod0; // Least significant 256 bits of the product
            uint256 prod1; // Most significant 256 bits of the product
            assembly {
                let mm := mulmod(x, y, not(0))
                prod0 := mul(x, y)
                prod1 := sub(sub(mm, prod0), lt(mm, prod0))
            }

            // Handle non-overflow cases, 256 by 256 division.
            if (prod1 == 0) {
                return prod0 / denominator;
            }

            // Make sure the result is less than 2^256. Also prevents denominator == 0.
            require(denominator > prod1);

            ///////////////////////////////////////////////
            // 512 by 256 division.
            ///////////////////////////////////////////////

            // Make division exact by subtracting the remainder from [prod1 prod0].
            uint256 remainder;
            assembly {
                // Compute remainder using mulmod.
                remainder := mulmod(x, y, denominator)

                // Subtract 256 bit number from 512 bit number.
                prod1 := sub(prod1, gt(remainder, prod0))
                prod0 := sub(prod0, remainder)
            }

            // Factor powers of two out of denominator and compute largest power of two divisor of denominator. Always >= 1.
            // See https://cs.stackexchange.com/q/138556/92363.

            // Does not overflow because the denominator cannot be zero at this stage in the function.
            uint256 twos = denominator & (~denominator + 1);
            assembly {
                // Divide denominator by twos.
                denominator := div(denominator, twos)

                // Divide [prod1 prod0] by twos.
                prod0 := div(prod0, twos)

                // Flip twos such that it is 2^256 / twos. If twos is zero, then it becomes one.
                twos := add(div(sub(0, twos), twos), 1)
            }

            // Shift in bits from prod1 into prod0.
            prod0 |= prod1 * twos;

            // Invert denominator mod 2^256. Now that denominator is an odd number, it has an inverse modulo 2^256 such
            // that denominator * inv = 1 mod 2^256. Compute the inverse by starting with a seed that is correct for
            // four bits. That is, denominator * inv = 1 mod 2^4.
            uint256 inverse = (3 * denominator) ^ 2;

            // Use the Newton-Raphson iteration to improve the precision. Thanks to Hensel's lifting lemma, this also works
            // in modular arithmetic, doubling the correct bits in each step.
            inverse *= 2 - denominator * inverse; // inverse mod 2^8
            inverse *= 2 - denominator * inverse; // inverse mod 2^16
            inverse *= 2 - denominator * inverse; // inverse mod 2^32
            inverse *= 2 - denominator * inverse; // inverse mod 2^64
            inverse *= 2 - denominator * inverse; // inverse mod 2^128
            inverse *= 2 - denominator * inverse; // inverse mod 2^256

            // Because the division is now exact we can divide by multiplying with the modular inverse of denominator.
            // This will give us the correct result modulo 2^256. Since the preconditions guarantee that the outcome is
            // less than 2^256, this is the final result. We don't need to compute the high bits of the result and prod1
            // is no longer required.
            result = prod0 * inverse;
            return result;
        }
    }

    /**
     * @notice Calculates x * y / denominator with full precision, following the selected rounding direction.
     */
    function mulDiv(
        uint256 x,
        uint256 y,
        uint256 denominator,
        Rounding rounding
    ) internal pure returns (uint256) {
        uint256 result = mulDiv(x, y, denominator);
        if (rounding == Rounding.Up && mulmod(x, y, denominator) > 0) {
            result += 1;
        }
        return result;
    }

    /**
     * @dev Returns the square root of a number. If the number is not a perfect square, the value is rounded down.
     *
     * Inspired by Henry S. Warren, Jr.'s "Hacker's Delight" (Chapter 11).
     */
    function sqrt(uint256 a) internal pure returns (uint256) {
        if (a == 0) {
            return 0;
        }

        // For our first guess, we get the biggest power of 2 which is smaller than the square root of the target.
        //
        // We know that the "msb" (most significant bit) of our target number `a` is a power of 2 such that we have
        // `msb(a) <= a < 2*msb(a)`. This value can be written `msb(a)=2**k` with `k=log2(a)`.
        //
        // This can be rewritten `2**log2(a) <= a < 2**(log2(a) + 1)`
        // → `sqrt(2**k) <= sqrt(a) < sqrt(2**(k+1))`
        // → `2**(k/2) <= sqrt(a) < 2**((k+1)/2) <= 2**(k/2 + 1)`
        //
        // Consequently, `2**(log2(a) / 2)` is a good first approximation of `sqrt(a)` with at least 1 correct bit.
        uint256 result = 1 << (log2(a) >> 1);

        // At this point `result` is an estimation with one bit of precision. We know the true value is a uint128,
        // since it is the square root of a uint256. Newton's method converges quadratically (precision doubles at
        // every iteration). We thus need at most 7 iteration to turn our partial result with one bit of precision
        // into the expected uint128 result.
        unchecked {
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            result = (result + a / result) >> 1;
            return min(result, a / result);
        }
    }

    /**
     * @notice Calculates sqrt(a), following the selected rounding direction.
     */
    function sqrt(uint256 a, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = sqrt(a);
            return result + (rounding == Rounding.Up && result * result < a ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 2, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 128;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 64;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 32;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 16;
            }
            if (value >> 8 > 0) {
                value >>= 8;
                result += 8;
            }
            if (value >> 4 > 0) {
                value >>= 4;
                result += 4;
            }
            if (value >> 2 > 0) {
                value >>= 2;
                result += 2;
            }
            if (value >> 1 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 2, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log2(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log2(value);
            return result + (rounding == Rounding.Up && 1 << result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 10, rounded down, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >= 10**64) {
                value /= 10**64;
                result += 64;
            }
            if (value >= 10**32) {
                value /= 10**32;
                result += 32;
            }
            if (value >= 10**16) {
                value /= 10**16;
                result += 16;
            }
            if (value >= 10**8) {
                value /= 10**8;
                result += 8;
            }
            if (value >= 10**4) {
                value /= 10**4;
                result += 4;
            }
            if (value >= 10**2) {
                value /= 10**2;
                result += 2;
            }
            if (value >= 10**1) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log10(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log10(value);
            return result + (rounding == Rounding.Up && 10**result < value ? 1 : 0);
        }
    }

    /**
     * @dev Return the log in base 256, rounded down, of a positive value.
     * Returns 0 if given 0.
     *
     * Adding one to the result gives the number of pairs of hex symbols needed to represent `value` as a hex string.
     */
    function log256(uint256 value) internal pure returns (uint256) {
        uint256 result = 0;
        unchecked {
            if (value >> 128 > 0) {
                value >>= 128;
                result += 16;
            }
            if (value >> 64 > 0) {
                value >>= 64;
                result += 8;
            }
            if (value >> 32 > 0) {
                value >>= 32;
                result += 4;
            }
            if (value >> 16 > 0) {
                value >>= 16;
                result += 2;
            }
            if (value >> 8 > 0) {
                result += 1;
            }
        }
        return result;
    }

    /**
     * @dev Return the log in base 10, following the selected rounding direction, of a positive value.
     * Returns 0 if given 0.
     */
    function log256(uint256 value, Rounding rounding) internal pure returns (uint256) {
        unchecked {
            uint256 result = log256(value);
            return result + (rounding == Rounding.Up && 1 << (result * 8) < value ? 1 : 0);
        }
    }
}

// File: @openzeppelin/contracts/utils/Strings.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Strings.sol)

pragma solidity ^0.8.0;

import "./math/Math.sol";

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant _SYMBOLS = "0123456789abcdef";
    uint8 private constant _ADDRESS_LENGTH = 20;

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        unchecked {
            uint256 length = Math.log10(value) + 1;
            string memory buffer = new string(length);
            uint256 ptr;
            /// @solidity memory-safe-assembly
            assembly {
                ptr := add(buffer, add(32, length))
            }
            while (true) {
                ptr--;
                /// @solidity memory-safe-assembly
                assembly {
                    mstore8(ptr, byte(mod(value, 10), _SYMBOLS))
                }
                value /= 10;
                if (value == 0) break;
            }
            return buffer;
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        unchecked {
            return toHexString(value, Math.log256(value) + 1);
        }
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = _SYMBOLS[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

    /**
     * @dev Converts an `address` with fixed length of 20 bytes to its not checksummed ASCII `string` hexadecimal representation.
     */
    function toHexString(address addr) internal pure returns (string memory) {
        return toHexString(uint256(uint160(addr)), _ADDRESS_LENGTH);
    }
}

// File: contracts/Builder.sol



// SPDX-License-Identifier: GPL-3.0

pragma solidity ^0.8.15;
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/utils/Strings.sol";
import "erc721a-upgradeable/contracts/interfaces/IERC721AUpgradeable.sol";

contract Builder is Ownable {
    using Strings for uint256;
    using Strings for uint160;

    string[] internal titles1;
    string[] internal titles2;
    string[] internal activities;
    string[] internal materials;
    string[] internal observations;
    string[] internal symbols;
    string[] internal metaphors;
    string[] internal amounts;
    string[] internal objectsGroup1; // strings
    string[] internal objectsGroup2; // 'a [color] [object]'
    string[] internal objectsGroup3; // '[amount] [object]'
    string[] internal objectsGroup4; // '[amount] [objectColor] [object]'
    string[] internal objectsGroup5Prefixes; // '[num] [object]'
    string[] internal objectsGroup5Suffixes; // 'a set of/a sculpture made of [color] dumbells/Play-doh'
    string[] internal domains;
    string[] internal domains2;

    // Metadata for vars
    string[] internal materialsMeta;
    string[] internal activitiesMeta;
    string[] internal symbolsMeta;
    string[] internal metaphorsMeta;
    string[] internal domainsMeta;
    string[] internal objectsMeta;
    string[] internal observationsMeta;

    IERC721AUpgradeable public tokenContract;

    function constructTokenDNA(
        uint256 tokenId
    ) external view returns (uint256[9] memory tokenDNA) {
        require(
            tokenContract.totalSupply() >= tokenId,
            "This token doesn't exist."
        );
        tokenDNA = [
            getTraitIndex(tokenId, "TITLE1", titles1.length),
            getTraitIndex(tokenId, "TITLE2", titles2.length),
            getTraitIndex(tokenId, "ACTIVITY", activities.length),
            getTraitIndex(tokenId, "MATERIALS", materials.length),
            getTraitIndex(
                tokenId,
                "OBJECTS",
                (objectsGroup1.length +
                    objectsGroup2.length +
                    objectsGroup3.length +
                    objectsGroup4.length +
                    objectsGroup5Prefixes.length)
            ),
            getTraitIndex(tokenId, "DOMAINS", domains.length),
            getTraitIndex(tokenId, "OBSERVATIONS", observations.length),
            getTraitIndex(tokenId, "SYMBOLS", symbols.length),
            getTraitIndex(tokenId, "METAPHORS", metaphors.length)
        ];
    }

    function getTokenIdSvg(
        string memory title,
        string memory s1,
        string memory s2
    ) external view returns (string memory svg) {
        string[3] memory sentenceLines = this.wordwrap(bytes(s1));
        string[3] memory sentenceLines2 = this.wordwrap(bytes(s2));

        return
            string(
                abi.encodePacked(
                    '<svg xmlns="http://www.w3.org/2000/svg" preserveAspectRatio="xMinYMin meet" width="2000" height="2000" viewBox="0 0 350 350"><style>.base{font-family:Helvetica,serif;font-size:14px}</style><rect width="100%" height="100%"/><text x="20" y="70" class="base" fill="#FFF" style="font-weight:700;font-size:18px">',
                    title,
                    '</text><text x="20" y="140" class="base" fill="#FFF">',
                    sentenceLines[0],
                    '</text><text x="20" y="160" class="base" fill="#FFF"> ',
                    sentenceLines[1],
                    '</text><text x="20" y="180" class="base" fill="#FFF"> ',
                    sentenceLines[2],
                    '</text><text x="20" y="260" class="base" fill="#FFF">',
                    sentenceLines2[0],
                    '</text><text x="20" y="280" class="base" fill="#FFF">',
                    sentenceLines2[1],
                    '</text><text x="20" y="300" class="base" fill="#FFF">',
                    sentenceLines2[2],
                    "</text></svg>"
                )
            );
    }

    function getSentence1(
        uint256 tokenId,
        uint256[9] memory tokenDNA
    ) external view returns (string memory sentence) {
        return
            string(
                abi.encodePacked(
                    "An artist ",
                    activities[tokenDNA[2]],
                    " ",
                    getObject(tokenId, tokenDNA[4], tokenDNA[3]),
                    " ",
                    domains[tokenDNA[5]],
                    "."
                )
            );
    }

    function getSentence2(
        uint256[9] memory tokenDNA,
        string memory title
    ) external view returns (string memory sentence) {
        return
            string(
                abi.encodePacked(
                    title,
                    " ",
                    observations[tokenDNA[6]],
                    " ",
                    symbols[tokenDNA[7]],
                    " ",
                    "as a ",
                    metaphors[tokenDNA[8]],
                    "."
                )
            );
    }

    function getObject(
        uint256 tokenId,
        uint256 index,
        uint256 materialIndex
    ) internal view returns (string memory) {
        uint256 rangeIndex;
        uint256 rand = uint256(keccak256(abi.encodePacked(tokenId + index)));
        string memory object = index.toString();
        uint256[5] memory ranges = getObjectRanges();

        if (index < ranges[0]) {
            // standard string
            object = objectsGroup1[index];
        }
        if (index >= ranges[0] && index < ranges[1]) {
            // pattern: 'a [color] [object]'
            string memory color = materials[materialIndex];
            rangeIndex = index - ranges[0];
            object = string(
                abi.encodePacked("a ", color, objectsGroup2[rangeIndex])
            );
        }
        if (index >= ranges[1] && index < ranges[2]) {
            // pattern: '[amount] [object]'
            rangeIndex = index - ranges[1];
            string memory amount = amounts[rand % amounts.length];
            object = string(
                abi.encodePacked(amount, objectsGroup3[rangeIndex])
            );
        }
        if (index >= ranges[2] && index < ranges[3]) {
            // pattern: '[amount] [color] [object]'
            rangeIndex = index - ranges[2];
            string memory amount = amounts[rand % amounts.length];
            string memory color = materials[materialIndex];
            object = string(
                abi.encodePacked(amount, " ", color, objectsGroup4[rangeIndex])
            );
        }
        if (index >= ranges[3] && index < ranges[4]) {
            // pattern: 'a set of/a sculpture made of [color] dumbells/Play-doh'
            rangeIndex = index - ranges[3];
            string memory color = materials[materialIndex];
            object = string(
                abi.encodePacked(
                    objectsGroup5Prefixes[rangeIndex],
                    " ",
                    color,
                    objectsGroup5Suffixes[rangeIndex]
                )
            );
        }
        return object;
    }

    function getTitle(
        uint256 titles1Index,
        uint256 titles2Index
    ) external view returns (string memory) {
        string memory title = string(
            abi.encodePacked(titles1[titles1Index], " ", titles2[titles2Index])
        );
        return title;
    }

    /*
    ⌐◨—————————————————————————————————————————————————————————————◨
                           utility functions
    ⌐◨—————————————————————————————————————————————————————————————◨ 
    */

    function getTraitIndex(
        uint256 tokenId,
        string memory keyPrefix,
        uint256 max
    ) internal pure returns (uint256 index) {
        uint256 rand = randomNum(
            string(abi.encodePacked(keyPrefix, tokenId.toString()))
        ) % (max - 1);

        index = rand;
    }

    function randomNum(string memory input) internal pure returns (uint256) {
        return uint256(keccak256(abi.encodePacked(input)));
    }

    function wordwrap(
        bytes calldata descBytes
    ) public pure returns (string[3] memory) {
        string[3] memory split = ["", "", ""];
        uint8 maxLength = 45;
        uint8 firstSplit = maxLength - 1;

        if (descBytes.length <= firstSplit) {
            split[0] = string(descBytes); // Short enough to go on one line
        } else {
            // look for place to make the first linebreak
            while (descBytes[firstSplit] != " ") {
                firstSplit--;
            }
            if (descBytes[firstSplit] == " ") {
                split[0] = string(descBytes[:firstSplit]);
                split[1] = string(descBytes[firstSplit:]);
            }
            if (bytes(split[1]).length > maxLength) {
                uint8 secondSplit = firstSplit + maxLength;
                while (descBytes[secondSplit] != " ") {
                    secondSplit--;
                }
                split[1] = string(descBytes[firstSplit:secondSplit]);
                split[2] = string(descBytes[secondSplit:]);
            }
        }

        return split;
    }

    function getMetadataValue(
        uint256 traitIndex,
        uint256 DNAIndex
    ) external view returns (string memory value) {
        if (DNAIndex == 2) {
            value = activitiesMeta[traitIndex];
        }
        if (DNAIndex == 3) {
            value = materialsMeta[traitIndex];
        }
        if (DNAIndex == 4) {
            value = objectsMeta[traitIndex];
        }
        if (DNAIndex == 5) {
            value = domainsMeta[traitIndex];
        }
        if (DNAIndex == 6) {
            value = observationsMeta[traitIndex];
        }
        if (DNAIndex == 7) {
            value = symbolsMeta[traitIndex];
        }
        if (DNAIndex == 8) {
            value = metaphorsMeta[traitIndex];
        }
    }

    function getObjectRanges() public view returns (uint256[5] memory) {
        uint256[5] memory objectRanges = [
            objectsGroup1.length,
            objectsGroup1.length + objectsGroup2.length,
            objectsGroup1.length + objectsGroup2.length + objectsGroup3.length,
            objectsGroup1.length +
                objectsGroup2.length +
                objectsGroup3.length +
                objectsGroup4.length,
            objectsGroup1.length +
                objectsGroup2.length +
                objectsGroup3.length +
                objectsGroup4.length +
                objectsGroup5Prefixes.length
        ];
        return objectRanges;
    }

    function setTokenContract(
        IERC721AUpgradeable _tokenContract
    ) external onlyOwner {
        tokenContract = _tokenContract;
    }

    function setTitles1(string[] memory _titles1) external onlyOwner {
        titles1 = _titles1;
    }

    function setTitles2(string[] memory _titles2) external onlyOwner {
        titles2 = _titles2;
    }

    function setActivities(string[] memory _activities) external onlyOwner {
        activities = _activities;
    }

    function setMaterials(string[] memory _materials) external onlyOwner {
        materials = _materials;
    }

    function setObservations(string[] memory _observations) external onlyOwner {
        observations = _observations;
    }

    function setSymbols(string[] memory _symbols) external onlyOwner {
        symbols = _symbols;
    }

    function setMetaphors(string[] memory _metaphors) external onlyOwner {
        metaphors = _metaphors;
    }

    function setAmounts(string[] memory _amounts) external onlyOwner {
        amounts = _amounts;
    }

    function setObjectsGroup1(string[] memory _objects) external onlyOwner {
        objectsGroup1 = _objects;
    }

    function setObjectsGroup2(string[] memory _objects) external onlyOwner {
        objectsGroup2 = _objects;
    }

    function setObjectsGroup3(string[] memory _objects) external onlyOwner {
        objectsGroup3 = _objects;
    }

    function setObjectsGroup4(string[] memory _objects) external onlyOwner {
        objectsGroup4 = _objects;
    }

    function setObjectsGroup5Prefixes(
        string[] memory _objectsPrefix
    ) external onlyOwner {
        objectsGroup5Prefixes = _objectsPrefix;
    }

    function setObjectsGroup5Suffixes(
        string[] memory _objectsSuffix
    ) external onlyOwner {
        objectsGroup5Suffixes = _objectsSuffix;
    }

    function setDomains(string[] memory _domains) external onlyOwner {
        domains = _domains;
    }

    /*
    ⌐◨—————————————————————————————————————————————————————————————◨
                           metadata
    ⌐◨—————————————————————————————————————————————————————————————◨ 
    */

    function setMaterialsMeta(
        string[] memory _materialsMeta
    ) external onlyOwner {
        materialsMeta = _materialsMeta;
    }

    function setActivitiesMeta(
        string[] memory _activitiesMeta
    ) external onlyOwner {
        activitiesMeta = _activitiesMeta;
    }

    function setSymbolsMeta(string[] memory _symbolsMeta) external onlyOwner {
        symbolsMeta = _symbolsMeta;
    }

    function setMetaphorsMeta(
        string[] memory _metaphorsMeta
    ) external onlyOwner {
        metaphorsMeta = _metaphorsMeta;
    }

    function setDomainsMeta(string[] memory _domainsMeta) external onlyOwner {
        domainsMeta = _domainsMeta;
    }

    function setObjectsMeta(string[] memory _objectsMeta) external onlyOwner {
        objectsMeta = _objectsMeta;
    }

    function setObservationsMeta(
        string[] memory _observationsMeta
    ) external onlyOwner {
        observationsMeta = _observationsMeta;
    }
}

// File: erc721a-upgradeable/contracts/IERC721AUpgradeable.sol



// SPDX-License-Identifier: MIT
// ERC721A Contracts v4.2.3
// Creator: Chiru Labs

pragma solidity ^0.8.4;

/**
 * @dev Interface of ERC721A.
 */
interface IERC721AUpgradeable {
    /**
     * The caller must own the token or be an approved operator.
     */
    error ApprovalCallerNotOwnerNorApproved();

    /**
     * The token does not exist.
     */
    error ApprovalQueryForNonexistentToken();

    /**
     * Cannot query the balance for the zero address.
     */
    error BalanceQueryForZeroAddress();

    /**
     * Cannot mint to the zero address.
     */
    error MintToZeroAddress();

    /**
     * The quantity of tokens minted must be more than zero.
     */
    error MintZeroQuantity();

    /**
     * The token does not exist.
     */
    error OwnerQueryForNonexistentToken();

    /**
     * The caller must own the token or be an approved operator.
     */
    error TransferCallerNotOwnerNorApproved();

    /**
     * The token must be owned by `from`.
     */
    error TransferFromIncorrectOwner();

    /**
     * Cannot safely transfer to a contract that does not implement the
     * ERC721Receiver interface.
     */
    error TransferToNonERC721ReceiverImplementer();

    /**
     * Cannot transfer to the zero address.
     */
    error TransferToZeroAddress();

    /**
     * The token does not exist.
     */
    error URIQueryForNonexistentToken();

    /**
     * The `quantity` minted with ERC2309 exceeds the safety limit.
     */
    error MintERC2309QuantityExceedsLimit();

    /**
     * The `extraData` cannot be set on an unintialized ownership slot.
     */
    error OwnershipNotInitializedForExtraData();

    // =============================================================
    //                            STRUCTS
    // =============================================================

    struct TokenOwnership {
        // The address of the owner.
        address addr;
        // Stores the start time of ownership with minimal overhead for tokenomics.
        uint64 startTimestamp;
        // Whether the token has been burned.
        bool burned;
        // Arbitrary data similar to `startTimestamp` that can be set via {_extraData}.
        uint24 extraData;
    }

    // =============================================================
    //                         TOKEN COUNTERS
    // =============================================================

    /**
     * @dev Returns the total number of tokens in existence.
     * Burned tokens will reduce the count.
     * To get the total number of tokens minted, please see {_totalMinted}.
     */
    function totalSupply() external view returns (uint256);

    // =============================================================
    //                            IERC165
    // =============================================================

    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * [EIP section](https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified)
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);

    // =============================================================
    //                            IERC721
    // =============================================================

    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables
     * (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in `owner`'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`,
     * checking first that contract recipients are aware of the ERC721 protocol
     * to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move
     * this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement
     * {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId,
        bytes calldata data
    ) external payable;

    /**
     * @dev Equivalent to `safeTransferFrom(from, to, tokenId, '')`.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external payable;

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom}
     * whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token
     * by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 tokenId
    ) external payable;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the
     * zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external payable;

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom}
     * for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}.
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    // =============================================================
    //                        IERC721Metadata
    // =============================================================

    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);

    // =============================================================
    //                           IERC2309
    // =============================================================

    /**
     * @dev Emitted when tokens in `fromTokenId` to `toTokenId`
     * (inclusive) is transferred from `from` to `to`, as defined in the
     * [ERC2309](https://eips.ethereum.org/EIPS/eip-2309) standard.
     *
     * See {_mintERC2309} for more details.
     */
    event ConsecutiveTransfer(uint256 indexed fromTokenId, uint256 toTokenId, address indexed from, address indexed to);
}

// File: erc721a-upgradeable/contracts/interfaces/IERC721AUpgradeable.sol



// SPDX-License-Identifier: MIT
// ERC721A Contracts v4.2.3
// Creator: Chiru Labs

pragma solidity ^0.8.4;

import '../IERC721AUpgradeable.sol';